
Courbes de poursuites avec Scilab 
 

Fakhreddine GHOMMID, lycée Gustave Eiffel, Dijon 
 
 
Résumé : Exemple d’utilisation du logiciel Scilab dans la résolution approchée d’un 
problème historique, en l’occurrence, le problème des courbes de poursuite. 
 
Mots clés : Courbe de poursuite ; logiciel Scilab ; représentation des courbes en 
informatique ; motivation à la recherche. 
 
 

Les activités d’algorithmique développées dans les classes de lycées, 
permettent de revisiter de beaux problèmes, désormais devenus des « classiques », en 
ajoutant un caractère parfois ludique facilitant l’exploration et la recherche.  

Le logiciel de calcul scientifique Scilab, libre et gratuit, est ici mis en avant 
afin de montrer quelques-unes de ses possibilités ; l’activité choisie pour cela est celle 
de la construction d’une courbe de poursuite. 

Mais avant d’aller plus loin, donnons la définition de ces courbes que proposa 
Bouguer, publiée dans les Mémoires de mathématiques et de physique de l’Académie 
Royale des Sciences, 1732 : 

 
[…] On peut donner à ces courbes 

le nom de lignes de poursuites, parce 
qu’elles sont décrites par le mouvement 
uniforme d’un point qui en poursuit un 
autre. Considérées sous cette idée, on 
voit qu’elles sont tracées presque tous 
les jours, et elles l’étaient le temps passé 
encore plus souvent, lorsqu’on 
commettait fréquemment cette faute dans 
la Marine, de diriger exactement la 
Proue vers les Vaisseaux auxquels on 
donnait chasse. Aussitôt que le Navire 
qui poursuivait l’autre, n’était pas situé 
sur la route que suivait ce dernier, il 
était obligé de se détourner dans celle de 
la ligne droite, et de décrire la courbe 
dont il est ici question. 

Examinons à l’aide de schémas l’exemple proposé : 
 
Dans le plan, le point V représente la 
position du vaisseau poursuivi (V0 étant 
sa position initiale) et N représente la 
position du navire prenant le vaisseau en 
chasse (N0 étant sa position initiale). V 
décrit une demi-droite et N la courbe de 
poursuite Cp, la droite (NV) étant 

supposée tangente à la courbe Cp. 

 

 
Afin de modéliser approximativement cette dynamique à l’aide d’un algorithme, 
donnons-nous une approche discrète, une construction « pas-à-pas » par intervalles de 
temps : 
 
 
 
 
 
 
 
 
Suivant cette approximation, et étant données deux positions N1 et V1, les nouvelles 
positions N2 et V2 seront, un laps de temps après, telles que 1 2 1 1 2N N k N V

 
 et 

1 2 2VV k u
 

 où k1 et k2 sont deux constantes réelles à déterminer et u


 un vecteur 
directeur de la trajectoire suivie par le vaisseau. 
Mais essayons de construire point par point, « à la main » (en fait ici avec Geogebra), 
les positions successives de nos deux bateaux. Remarquons qu’entre les instants 
successifs t=0, , 2, 3,… les distances parcourues restent constantes : 

Ces constructions successives esquissent notre premier algorithme dont le code 
Scilab est donné ci-après : 

Code Scilab 
function result=distance(A, B) 
    result=sqrt((B(1)-A(1))^2+(B(2)-A(2))^2); 
endfunction 
 
function [coords]=deplace(A, u) 
    coords(1)=A(1)+u(1) 
    coords(2)=A(2)+u(2) 
endfunction 
 
function [coords]=vecteur(A, B) 
    coords(1)=B(1)-A(1) 
    coords(2)=B(2)-A(2) 
endfunction 
 
d1=1.5/10    //Déplacement du Navire par intervalle de temps 
d2=1/10       //Déplacement du Vaisseau par intervalle de temps 
 
N=[-5,-2]      //Position initiale du Navire (coordonnées) 
V=[-4,0.5]    //Position initiale du Vaisseau (coordonnées) 
  
clf()                //Efface le contenu de la fenêtre graphique 
orthonorme    //Rend le repère orthonormé 
compteur=1 
while ((distance(N,V)>0.1)&(compteur<100)) 
    compteur=compteur+1 
    A=N    //Mémorise les anciennes positions de N et V 
    B=V    //pour le tracé des segments avec plot2d 
    k1=d1/distance(N,V) 
    N=deplace(N,vecteur(N,V).*k1)    //Nouvelles coordonnées du Navire 
    V=deplace(V,[d2,0])             //Nouvelles coordonnées du Vaisseau 
    plot2d([A(1),N(1)],[A(2),N(2)],[1],axesflag=0,rect=[-5,-2,2,1]) 
    plot2d([B(1),V(1)],[B(2),V(2)],[5],axesflag=0,rect=[-5,-2,2,1]) 
end 

Exemples de sorties 

pour différentes valeurs de d1, d2, N et V. 
(précisées et modifiables dans l’algorithme) 

 
d1=1.5/10     
d2=1/10        
N=[-5,-2]      
V=[-4,0.5]     

 
 
d1=1.4/10     
d2=0.9/10        
N=[-4,-2]      
V=[-4,0.5] 

 
d1=1.2/10     
d2=1/10        
N=[0,-3]      
V=[-4,0.5] 

 
Ce type de programme peut être introduit très tôt au lycée, dès les développements 
des premiers algorithmes relatifs à la géométrie repérée. On pourrait bien sûr utiliser 
le très ludique logiciel Scratch, mais le contenu mathématique, déjà présenté a 
minima, serait vidé de toute substance : 
 
Code Lutin 2 

 
 

 
Code Lutin  

Affichage 
 

 
 
Une autre façon de traduire cette situation de poursuite est d’introduire les suites de 
coordonnées dans un repère orthonormé du plan d’origine V0, l’axe des abscisses 

étant dirigé et orienté par le parcours du vaisseau. En notant   
 et KK

tout entier naturel k, les positions successives de N et V, la relation vectorielle 

 (où d1 et d2 sont toujours les distances 

parcourues par intervalle de temps donné), conduit à : 

La représentation des suites avec Scilab étant faite sous la forme d’un tableau, 
l’indice initial ne peut être 0, mais 1 ; d’où une nécessaire adaptation que 
voici (calcul de 100 points pour l’exemple) : 

Code Scilab 

d1=1.5/10    //Déplacement du Navire par intervalle de temps 
d2=1/10      //Déplacement du Vaisseau par intervalle de temps 
clf() 
x(1)=-1    //Position initiale du Navire 
y(1)=-2 
for n=1:100 
    c=d1/sqrt((d2*(n-1)-x(n))^2+y(n)^2) 
    x(n+1)=c*(d2*(n-1)-x(n))+x(n) 
    y(n+1)=y(n)*(1-c)     
end 
plot2d(x,y,[1],axesflag=0,rect=[-5,-2,2,1]) 
 
Graphique obtenu 
 

 
 

 
Enfin, les possibilités de résolution numériques d’équations différentielles par Scilab 
peuvent être exploitées. 

En considérant le repère précédent avec 

, où x et y sont deux 

fonctions supposées dérivables de la variable réelle t, v1 et v2 les vitesses respectives 

du Vaisseau et du Navire, nous avons la relation 2dON v
NV

dt NV


 
 qui se traduit par le 

système différentiel suivant : 

Qui peut être résolu de façon approchée par Scilab comme suit : 

Code Scilab 

//Définition du système différentiel : 
function [f]=equadiff(t, N) 
    // Ici, f(1)=x'(t), f(2)=y'(t) et N=[x(t),y(t)] 
    f(1)=v2/sqrt((v1*t-N(1))^2+N(2)^2)*(v1*t-N(1)) 
    f(2)=-v2/sqrt((v1*t-N(1))^2+N(2)^2)*N(2) 
endfunction 
 
//Valeurs des vitesses respectives : 
v1=1 
v2=2 
clf() 
// les valeurs des instants pour lesquels on récupère la solution (tmin=0, 
tmax=500) 
t = linspace(0,1,500)  
N0 = [1 ; -1];         // la condition initiale 
// La résolution approchée de notre équation différentielle : 
[N] = ode(N0, 0, t, equadiff)  
// Le tracé de la solution (sous réserve d'unicité): 
plot2d(N(1,:)',N(2,:)') 
 
Exemples de courbes obtenues 

Les différentes approches 
présentées ne doivent pas faire 
oublier qu’aucune technique 
mobilisée ici ne valide les résultats 
obtenus (erreur liée à la méthode et 
stabilité des solutions, erreur 
numérique, etc.). Pour comparer les 
constructions approchées à la 
trajectoire exacte, laissons à M. de 
Maupertuis le soin de nous exposer 
sa solution (voir l’encadré ci-contre, 
extrait du même Mémoires de 
mathématiques et de physique de 
l’Académie Royale des Sciences de 
1732).  
 
« PROBLÈME 1. Trouver la courbe 
de poursuite, c’est-à-dire, la courbe 
par laquelle un Vaisseau doit en 
poursuivre un autre qui s’enfuit par 
une ligne droite, en supposant que les vitesses des deux Vaisseaux soient 
toujours dans le même rapport. 
SOLUTION. Ce Problème se réduit à trouver la courbe où la resecte est 
proportionnelle à l’arc (j’appelle Resecte, la partie de l’abscisse prise depuis 
l’origine jusqu’à la rencontre de la tangente). Prenant donc x pour l’abscisse, y 

pour l’ordonnée, & s pour l’arc, on a ,
ydx

x ms
dy

   ou 

2

2

dy dx ydyddx ydxddy
dx mds

dy

 
  . Cette équation, si l’on fait dy constant, se 

réduit à 
2 2(

ddx mdy

V dy dx y



, ou (intégrant par logarith.) 

2 2( )l dx dy dx ldy mly mlA     , ou, passant aux nombres, 
2 2 m mdx dy dx y A dy   , ou 2 2 m mdy dx y A dy dx   , ou 

2 2 2 2 2 22m m m mdy dx y A dy y A dydx dx     , ou 2 2 2 ,m m m mdy A y dy A y dx    ou 

1 1
,

2 2
m m m mdx A y dy A y dy  

 
ou enfin 1 11 1

Ou dans le cas m=1, 11 1
.

x A yy Aly C    » 

Il aboutit à démontrer que la courbe de poursuite admet, dans le cas 2

1

1
v

m
v

  , une 

équation de la forme :  
1 11 1

2( 1) 2( 1)
m mx y y B

m m
    

 
 

où  et B sont deux constantes réelles à déterminer ( mA   dans le texte en 
vis-à-vis). 
En ajoutant, par exemple, les lignes suivantes (dans le cas : N0 (2,2), V0 (2,2), et 
m=2/3) aux codes précédents : 
 

Code Scilab 
function x=f(y) 
    m=2/3 
    a=.6575230441 
    b=1.794112548 
    x=1/(2*(m+1)*a)*y^(m+1)+a/(2*(m-1))*y^(-m+1)+b 
endfunction 
y=linspace(0,2,50) 
x=f(y) 
plot2d(x,y,[5],axesflag=0,rect=[0,-1,4,2]) 

 
Nous obtenons en sortie graphique : 

 
Graphique qui nous permet d’apprécier visuellement la qualité des approximations. 
 

Bien d’autres exemples d’utilisation de fonctionnalités avancées, telles que des 
animations, auraient pu être présentées ici. Mais celles présentées couvrent déjà une 
part importante des programmes de lycée, à l’exception des probabilités et 
statistiques qui feront l’objet d’un prochain article. 

Pour conclure, et pour le plaisir, un bon moyen de savoir si on a bien compris 
un algorithme ou assimilé un langage informatique est de le modifier : dès lors, quelle 

serait l’allure d’une courbe de poursuite si le vaisseau poursuivi admet une trajectoire 
circulaire ? 
 
 

 

 

 

  

 
Dans chacun de ces quatre cas, les courbes de poursuites ont été tracées pour 
différentes positions de départ du poursuivant et différents rapports de vitesses. 

